
restart;
with(plots):
with(DEtools):

Warning, the name changecoords has been redefined

Intro Lyapunov functions: Simple pendulum

>
  

Content, Purpose and Use

This worksheet provides procedures to generate pictures that combine solution
curves of the DE for a simple (nonlinear) pendulum with the energy function
(candidate Lyapunov function).

>
  

The main purpose is to provide powerful images to be used as backdrops for
class-discussions.
The 3D-images are best shown live --- wiggle w/ the mouse, but can also be
printed for overhead slides.
Most plot ranges, viewing angles have been tuned for the selected initial conditions.
Expect substantial rework when changing the DE or IC.

>
  

The implementation uses lots of map and zip commands -- and is NOT meant to be
easily read -- instead writing it was a personal MAPLE coding challenge. Particularly
nice are the map and zip commands that create several simultanous plots of solution
curves, as defined by numeric-dvolve-output-procedures, and list these to the
energy surface and the decay dV/dt, re-interpreted as a function of the state.

>
  

Updates and log of modifications.

none yet.

>
  

>
  

Undamped and damped pendulum: Lyapunov stability

The differential equation and phase portrait: undamped case

>
  

ude:=diff(theta(t),t)=omega(t),diff(omega(t),t)=-sin(theta(t)):ude;
uic1:=theta(0)=-3/2*Pi,omega(0)=2:uic1;
uic2:=theta(0)=0,omega(0)=2-1/10:uic2;
vars:=theta(t),omega(t):

>
  

dsolve({ude,uic1},{vars});

>
  

a:=3*Pi:
b:=4:
DEplot([ude],[vars],t=0..20,[[uic1],[uic2]],theta=-a..a,omega=-b..b,
        stepsize=.05,method=rkf45,
        title=`Undamped nonlinear pendulum`,
        color=red,linecolor=blue,arrows=small,thickness=3);

>
  

The differential equation and phase portrait: the damped case

>
  

beta:=1/8;

>
  

dde:=diff(theta(t),t)=omega(t),diff(omega(t),t)=-sin(theta(t))-beta*omega(t):dde;
dic1:=theta(0)=-5/2*Pi,omega(0)=3:dic1;
dic2:=theta(0)=0,omega(0)=2-1/10:dic2;
vars:=theta(t),omega(t):

>
  

a:=3*Pi:
b:=4:
DEplot([dde],[vars],t=0..20,[[dic1],[dic2]],theta=-a..a,omega=-b..b,
        stepsize=.05,method=rkf45,
        title=`Undamped nonlinear pendulum`,
        color=red,linecolor=blue,arrows=small,thickness=3);

>
  

The energy as a function of the state

>
  

V:=omega^2/2+(1-cos(theta));

>
  

a:=3*Pi;
b:=3;
display([
   plot3d(V,theta=-a..a,omega=-b..b,style=patchcontour,shading=zhue)
  ,plot3d(0,theta=-a..a,omega=-b..b,style=wireframe,color=cyan)
],axes=normal,orientation=[53,53],view=[-a..a,-b..b,0..3]);

lyap1.html https://math.la.asu.edu/~kawski/MAPLE/475/lyap11.html

3 of 7 10/11/17, 12:23 PM



>
  

a:=5/4*Pi;
b:=3;
display([
   plot3d(V,theta=-a..a,omega=-b..b,style=patchcontour,shading=zhue)
  ,plot3d(0,theta=-a..a,omega=-b..b,style=wireframe,color=cyan)
],axes=normal,orientation=[60,60],view=[-a..a,-b..b,0..5/2]);

>

>
  

Energy along trajectories as a function of time

>
  

>
  

Vt:=subs(theta=theta(t),omega=omega(t),V);

>
  

usol1:=dsolve({ude,uic1},{vars},type=numeric);
usol2:=dsolve({ude,uic2},{vars},type=numeric);
dsol1:=dsolve({dde,dic1},{vars},type=numeric);
dsol2:=dsolve({dde,dic2},{vars},type=numeric);

>
  

display(zip((sol,cc)->
plot(map(tt->subs(sol(tt/10),[t,Vt]),[$0..15]),color=cc,thickness=3),
[usol1,usol2,dsol1,dsol2],[black,blue,red,magenta]),
title=`Energy along trajectories as a function of time`,
titlefont=[TIMES,BOLD,16],
axes=frame,
labels=[`time t`,`energy V`]);

>
  

;

>
  

Trajectories lifted to the energy surface

>
  

>
  

> arange:=-Pi..5*Pi:

  brange:=-3..3:
drop:=-3:
display(
[
op(
zip((sol,cc)->
    op([
    spacecurve(map(tt->subs(sol(tt/10),[theta(t),omega(t),Vt]),[$0..200]),color=cc,thickness=3),
    spacecurve(map(tt->subs(sol(tt/10),[theta(t),omega(t),drop]),[$0..200]),color=cc,thickness=3)
    ]),
    [usol1,usol2,dsol1,dsol2],[black,blue,red,magenta])
   ),
plot3d(drop,arange,brange,style=wireframe,color=cyan,grid=[30,20]),
plot3d(V,theta=arange,omega=brange,style=wireframe,shading=zhue,grid=[50,40])
],
orientation=[-120,70],
view=[arange,brange,drop..8],
title=`Integral curves and their lifts to the energy surface`,
titlefont=[TIMES,BOLD,16],
axes=frame,
labels=[`theta`,`omega`,`energy V`]);

>
  

>
  

>
  

Energy decay rate along trajectories

>
  

DVt:=diff(Vt,t);
DVt:=subs(dde,diff(Vt,t));
DV:=subs(theta(t)=theta,omega(t)=omega,simplify(DVt));

>
  

arange:=-Pi..5*Pi:
brange:=-2..5/2:
drop:=-3/2:
display(
[
op(
zip((sol,cc)->
    op([
    spacecurve(map(tt->subs(sol(tt/10),[theta(t),omega(t),DVt]),[$0..200]),color=cc,thickness=3),
    spacecurve(map(tt->subs(sol(tt/10),[theta(t),omega(t),drop]),[$0..200]),color=cc,thickness=3)
    ]),
    [dsol1,dsol2],[red,magenta])
   ),
plot3d(0,arange,brange,style=wireframe,color=gray,grid=[30,20]),
plot3d(drop,arange,brange,style=wireframe,color=cyan,grid=[30,20]),
plot3d(DV,theta=arange,omega=brange,style=wireframe,color=green,grid=[50,40])
],
orientation=[-130,65],
view=[arange,brange,drop..0],
title=`The decay dV/dt is NOT strictly negative`,
titlefont=[TIMES,BOLD,16],
axes=frame,tickmarks=[3,3,3],
labels=[`theta`,`omega`,`decay DV/dt`]);

>
  

>
  

>
  

>
  

>
  

>

